# Complex conjugate series

I have a series like this one:

series = a[1][v](r-rp) + a[2][v](r-rp)^2 + O[r-rp]^3


where a[n][v] are complex coefficients of the series which are functions of v and (r-rp) is a real number. I want to Conjugate[] the series. However, Mathematica produces only (whatever I tried):

Conjugate[a[1][v](r-rp) + a[2][v](r-rp)^2 + O[r-rp]^3]


while I would like to obtain:

Conjugate[a[1][v]](r-rp) + Conjugate[a[2][v]](r-rp)^2 + O[r-rp]^3


I have tried e.g. Conjugate[ComplexExpand[series,{a}]] and variations with no good.

What should I do?

First, note the InputForm of the series expression:

series //InputForm


SeriesData[r, rp, {a[1][v], a[2][v]}, 1, 3, 1]

According to the documentation, and as you can see above, the 3rd argument consists of the coefficients of the series. So, to take the conjugate of a SeriesData object, we just need to map Conjugate at the right part:

conj = MapAt[Conjugate, series, 3];
conj //TeXForm


$(r-\operatorname{rp}) (a(1)(v))^*+(r-\operatorname{rp})^2 (a(2)(v))^*+O\left((r-\operatorname{rp})^3\right)$

Quick-and-dirty approach:

series = a[1][v] (r - rp) + a[2][v] (r - rp)^2 + O[r, rp]^3;
Assuming[r \[Element] Reals && rp \[Element] Reals,
Conjugate /@ series // Simplify
]

• Thank you for the answer. Could you explain to me, why for test = a+b Conjugate[test] returns Conjugate[a+b] while Conjugate /@ test returns Conjugate[a]+Conjugate[b]. – fales Apr 5 '18 at 16:29
• @fales have a look at Map. – AccidentalFourierTransform Apr 5 '18 at 16:36